

NAG Fortran Library Routine Document

F07MDF (SSYTRF/DSYTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07MDF (SSYTRF/DSYTRF) computes the Bunch–Kaufman factorization of a real symmetric indefinite matrix.

2 Specification

```

SUBROUTINE F07MDF(UPLO, N, A, LDA, IPIV, WORK, LWORK, INFO)
ENTRY      ssytrf (UPLO, N, A, LDA, IPIV, WORK, LWORK, INFO)
INTEGER    N, LDA, IPIV(*), LWORK, INFO
real     A(LDA,*), WORK(*)
CHARACTER*1 UPLO

```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine factorizes a real symmetric matrix A , using the Bunch–Kaufman diagonal pivoting method. A is factorized as either $A = PUDU^T P^T$ if UPLO = 'U', or $A = PLDL^T P^T$ if UPLO = 'L', where P is a permutation matrix, U (or L) is a unit upper (or lower) triangular matrix and D is a symmetric block diagonal matrix with 1 by 1 and 2 by 2 diagonal blocks; U (or L) has 2 by 2 unit diagonal blocks corresponding to the 2 by 2 blocks of D . Row and column interchanges are performed to ensure numerical stability while preserving symmetry.

This method is suitable for symmetric matrices which are not known to be positive-definite. If A is in fact positive-definite, no interchanges are performed and no 2 by 2 blocks occur in D .

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates whether the upper or lower triangular part of A is stored and how A is to be factorized, as follows:

if UPLO = 'U', the upper triangular part of A is stored and A is factorized as $PUDU^T P^T$, where U is upper triangular;

if UPLO = 'L', the lower triangular part of A is stored and A is factorized as $PLDL^T P^T$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n , the order of the matrix A .

Constraint: $N \geq 0$.

- 3: A(LDA,*) – *real* array Input/Output
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the n by n symmetric matrix A . If UPLO = 'U', the upper triangle of A must be stored and the elements of the array below the diagonal are not referenced; if UPLO = 'L', the lower triangle of A must be stored and the elements of the array above the diagonal are not referenced.
On exit: the upper or lower triangle of A is overwritten by details of the block diagonal matrix D and the multipliers used to obtain the factor U or L as specified by UPLO.
- 4: LDA – INTEGER Input
On entry: the first dimension of the array A as declared in the (sub)program from which F07MDF (SSYTRF/DSYTRF) is called.
Constraint: $LDA \geq \max(1, N)$.
- 5: IPIV(*) – INTEGER array Output
Note: the dimension of the array IPIV must be at least $\max(1, N)$.
On exit: details of the interchanges and the block structure of D .
 More precisely, if $IPIV(i) = k > 0$, d_{ii} is a 1 by 1 pivot block and the i th row and column of A were interchanged with the k th row and column.
 If UPLO = 'U' and $IPIV(i-1) = IPIV(i) = -l < 0$, $\begin{pmatrix} d_{i-1,i-1} & d_{i,i-1} \\ d_{i,i-1} & d_{ii} \end{pmatrix}$ is a 2 by 2 pivot block and the $(i-1)$ th row and column of A were interchanged with the l th row and column.
 If UPLO = 'L' and $IPIV(i) = IPIV(i+1) = -m < 0$, $\begin{pmatrix} d_{ii} & d_{i+1,i} \\ d_{i+1,i} & d_{i+1,i+1} \end{pmatrix}$ is a 2 by 2 pivot block and the $(i+1)$ th row and column of A were interchanged with the m th row and column.
- 6: WORK(*) – *real* array Workspace
Note: the dimension of the array WORK must be at least $\max(1, LWORK)$.
On exit: if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.
- 7: LWORK – INTEGER Input
On entry: the dimension of the array WORK as declared in the (sub)program from which F07MDF (SSYTRF/DSYTRF) is called, unless LWORK = -1, in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).
Suggested value: for optimum performance LWORK should be at least $N \times nb$, where nb is the **blocksize**.
Constraint: $LWORK \geq 1$ or $LWORK = -1$.
- 8: INFO – INTEGER Output
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO = i , d_{ii} is exactly zero. The factorization has been completed but the block diagonal matrix D is exactly singular, and division by zero will occur if it is subsequently used to solve a system of linear equations or to compute A^{-1} .

7 Accuracy

If UPLO = 'U', the computed factors U and D are the exact factors of a perturbed matrix $A + E$, where

$$|E| \leq c(n)\epsilon P|U||D||U^T|P^T,$$

$c(n)$ is a modest linear function of n , and ϵ is the *machine precision*. If UPLO = 'L', a similar statement holds for the computed factors L and D .

8 Further Comments

The elements of D overwrite the corresponding elements of A ; if D has 2 by 2 blocks, only the upper or lower triangle is stored, as specified by UPLO.

The unit diagonal elements of U or L and the 2 by 2 unit diagonal blocks are not stored. The remaining elements of U and L are stored in the corresponding columns of the array A, but additional row interchanges must be applied to recover U or L explicitly (this is seldom necessary). If IPIV(i) = i , for $i = 1, 2, \dots, n$ (as is the case when A is positive-definite), then U or L is stored explicitly (except for its unit diagonal elements which are equal to 1).

The total number of floating-point operations is approximately $\frac{1}{3}n^3$.

A call to this routine may be followed by calls to the routines:

F07MEF (SSYTRS/DSYTRS) to solve $AX = B$;

F07MGF (SSYCON/DSYCON) to estimate the condition number of A ;

F07MJF (SSYTRI/DSYTRI) to compute the inverse of A .

The complex analogues of this routine are F07MRF (CHETRF/ZHETRF) for Hermitian matrices and F07NRF (CSYTRF/ZSYTRF) for symmetric matrices.

9 Example

To compute the Bunch–Kaufman factorization of the matrix A , where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07MDF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, LDA, LWORK
PARAMETER       (NMAX=8,LDA=NMAX,LWORK=64*NMAX)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER       UPLO
*      .. Local Arrays ..
real           A(LDA,NMAX), WORK(LWORK)
```

```

      INTEGER          IPIV(NMAX)
*      .. External Subroutines ..
EXTERNAL          ssytrf, X04CAF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07MDF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF
*
*      Factorize A
*
      CALL ssytrf(UPLO,N,A,LDA,IPIV,WORK,LWORK,INFO)
*
      WRITE (NOUT,*)
*
*      Print details of factorization
*
      IFAIL = 0
      CALL X04CAF(UPLO,'Nonunit',N,N,A,LDA,
+              'Details of factorization',IFAIL)
*
*      Print pivot indices
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'IPIV'
      WRITE (NOUT,99999) (IPIV(I),I=1,N)
*
      IF (INFO.NE.0) WRITE (NOUT,*) 'The factor D is singular'
*
      END IF
      STOP
*
99999 FORMAT ((3X,7I11))
      END

```

9.2 Program Data

F07MDF Example Program Data

```

4          :Value of N
'L'       :Value of UPLO
2.07
3.87  -0.21
4.20  1.87  1.15
-1.15  0.63  2.06  -1.81  :End of matrix A

```

9.3 Program Results

F07MDF Example Program Results

Details of factorization

	1	2	3	4
1	2.0700			
2	4.2000	1.1500		
3	0.2230	0.8115	-2.5907	
4	0.6537	-0.5960	0.3031	0.4074

IPIV

	-3	-3	3	4
--	----	----	---	---